
set matastrict off 
mata: 
mata clear 
mata mlib create libbayes , dir(PERSONAL) replace 
/*--------------------------------------------  
* Single categorical observation  
* P = probabilities of each cell (sum to 1)  
* returns selected cell number  
*--------------------------------------------*/ 
real scalar rCategorical(real vector P) 
{ 
   real scalar k,u,h,sumP 
 
   k=length(P) 
   u=runiform(1,1) 
   h=1 
   sumP=P[1] 
   while( u > sumP & h < k ) { 
      h++ 
      sumP=sumP+P[h] 
   } 
   return(h) 
}    
mata mlib add libbayes rCategorical() 
/*-------------------------------------------- 
* Single Dirichlet distribution - parameter alpha 
*--------------------------------------------*/ 
real colvector rDirichlet(real vector alpha) 
{ 
   real scalar h,sP 
   real colvector P 
    
   h=length(alpha) 
   P=J(h,1,0) 
   sP=0 
   for(i=1;i<=h;i++) { 
      P[i]=rgamma(1,1,alpha[i],1) 
      sP=sP+P[i] 
   } 
   return(P/sP) 
} 
mata mlib add libbayes rDirichlet() 
/*-------------------------------------------- 
* Single random multivariate normal variable* M = mean 
* L = cholesky(V), V=variance matrix 
*--------------------------------------------*/ 
real rowvector rMNormal(real rowvector M,real matrix L) 
{ 
   return(M+L*rnormal(rows(L),1,0,1)) 
} 
mata mlib add libbayes rMNormal() 
/*-------------------------------------------- 
* Single Wishart matrix 
* L = cholesky(invsym(R)), where R = k*variance matrix 
* k = degree of freedom 
*--------------------------------------------*/ 
real matrix rWishart(real matrix L,real scalar k ) 
{    



    p = rows(L) 
    B = J(p,p,0) 
    for(i=2;i<=p;i++) { 
       for(j=1;j<i;j++) { 
          B[i,j] = rnormal(1,1,0,1) 
       } 
    } 
    for(i=1;i<=p;i++) B[i,i] = sqrt(rchi2(1,1,k-i+1)) 
    return(L*B*B'*L') 
} 
mata mlib add libbayes rWishart() 
mata mlib index 
end 


